Résolution d'une équation cubique réelle - le cas de 3 racines réelles

Le but de cette page de calcul est de montrer comment on peut résoudre une équation cubique en utilisant les nombres complexes, même si les solutions sont toutes réelles! Commençons par une équation explicite qu'on voudrait résoudre dans $\mathbf{R}$:

$$x^3 - 3x + 1 = 0.$$

Une petite étude de la fonction $y = x^3 - 3x + 1$ montre que cette équation possède bien 3 racines réelles. Voici le graphe de la fonction:

```
plot(x^3-3*x+1,(x,-3,3))
```



On commence par substituer $x = u + v$ dans l'expression $x^3 - 3x + 1$.

```
var('u','v')
(x^3-3*x+1).subs(x=u+v)
```

$$(u+v)^3 - 3u - 3v + 1$$

...et on développe...

```
_.expand()
```

$$u^3 + 3u^2v + 3uv^2 + v^3 - 3u - 3v + 1$$

En regardant de près, on voit que cette expression est égale à $u^3 + v^3 + (u+v)(3uv - 3) + 1$, et en effet:

```
(_-(u^3+v^3+(u+v)*(3*u*v-3)+1)).expand()
```

$$0$$

L'idée de génie de Tartaglia était de supposer que $3uv - 3 = 0$, i.e, que $uv = 1$ ou encore que $v = 1/u$. Cela simplifie considérablement l'expression $u^3 + v^3 + (u+v)(3uv - 3) + 1$. Substituons donc $v = 1/u$ dans cette expression:

```
(u^3 + 3*u^2*v + 3*u*v^2 + v^3 - 3*u - 3*v + 1).subs(v=1/u)
```

$$u^3 + \frac{1}{u^3} + 1$$

Multiplions par u^3:

```
(_*u^3).expand()
```

$$u^6 + u^3 + 1$$

C'est une expression quadratique en u^3! Pour résumer, si $x^3 - 3x + 1 = 0$ et si on pose $x = u + v$ et $v = 1/u$, on a u^3 est solution de l'équation quadratique $y^2 + y + 1 = 0$. Or, les solutions de cette derrnière sont:

```
y1=(-1+I*sqrt(3))/2; y2=(-1-I*sqrt(3))/2
```

comme on peut vérifier:

```
(y1^2+y1+1).expand()
```

$$0$$

```
(y2^2+y2+1).expand()
```

0

Si on détermine les 3 racines cubiques u_1, u_2, u_3 de y_1 dans $\mathbf{C}$, et on pose $v_i = 1/u_i$, alors les nombres complexes $x_i = u_i + v_i$ sont solution de l'équation $x^3 - 3x + 1 = 0$, pour $i = 1, 2, 3$. On verra que ces nombres complexes sont en fait réels! Notons que y_1 est égal à la racine cubique de l'unité $e^{2\pi i/3}$, en effet:

```
bool(y1==exp(2*pi*I/3))
```

True

Du coup, les racines cubiques de y_1 sont:

```
u1=exp(2*I*pi/9); u2=u1*exp(2*I*pi/3); u3=u1*exp(4*I*pi/3)
```

Posons donc $v_i = 1/u_i$:

```
v1=1/u1; v2=1/u2; v3=1/u3
```

et $x_i = u_i + v_i$:

```
x1=u1+v1; x2=u2+v2; x3=u3+v3
```

Vérifions que les nombres complexes x_1, x_2, x_3 sont bien solution de l'équation $x^3 - 3x + 1 = 0$:

```
(x^3-3*x+1).subs(x=x1)
```

$$\left(e^{\left(\frac{2}{9} i \pi\right)} + e^{\left(-\frac{2}{9} i \pi\right)}\right)^3 - 3\, e^{\left(\frac{2}{9} i \pi\right)} - 3\, e^{\left(-\frac{2}{9} i \pi\right)} + 1$$

```
_.expand()
```

$$e^{\left(\frac{2}{3} i \pi\right)} + e^{\left(-\frac{2}{3} i \pi\right)} + 1$$

```
_.simplify()
```

0

et de même pour x_2 et x_3:

```
(x^3-3*x+1).subs(x=x2).expand().simplify()
```

0

```
(x^3-3*x+1).subs(x=x3).expand().simplify()
```

0

Montrons que les nombres x_1, x_2, x_3 sont en fait réels:

```
x1
```

$$e^{\left(\frac{2}{9} i \pi\right)} + e^{\left(-\frac{2}{9} i \pi\right)}$$

```
x2
```

$$e^{\left(\frac{8}{9} i \pi\right)} + e^{\left(-\frac{8}{9} i \pi\right)}$$

```
x3
```

$$e^{\left(\frac{14}{9} i \pi\right)} + e^{\left(-\frac{14}{9} i \pi\right)}$$

Comme chacun de ces nombres est un nombre complexe plus son conjugué, ils sont bien réels. Ca peut encore se voir en calculant leur partie imaginaire:

```
x1.imag().simplify()
```

$$0$$

```
x2.imag().simplify()
```

$$0$$

```
x3.imag().simplify()
```

$$0$$

Ils sont donc égaux à leurs parties réelles:

```
x1=x1.real();x2=x2.real();x3=x3.real()
```

Les solutions de l'équation $x^3 - 3x + 1 = 0$ sont donc:

```
x1
```

$$2 \cos\left(\frac{2}{9} \pi\right)$$

```
x2
```

$$2 \cos\left(\frac{8}{9} \pi\right)$$

```
x3
```

$$2 \cos\left(\frac{14}{9} \pi\right)$$

Ce qui est remarquable, c'est que les solutions ne sont pas sous forme radicale réelle, i.e., aucune de ces nombres n'est une expression faisant n'intervenir que les symboles $+, -, x, /, \cdot^n, \sqrt[n]{\cdot}$ et des nombres réels.

Si on fait un dessin du graphe de la fonction $y = x^3 - 3x + 1$ et les points $(x_i, 0)$, pour $i = 1, 2, 3$, on voit bien que les nombres réels x_1, x_2, x_3 en sont racine:

```
G = plot(x^3-3*x+1,(x,-3,3)) + point((x1,0),size=50,color='red')
+ point((x2,0),size=50,color='green') +
point((x3,0),size=50,color='blue')
G.show()
```